
1 The Game Of Acro-Nim

Acro-Nim (from the Latin acro, heights, hence “High Nim”) is a Nim variant with some interesting rules.

The basic game is 1-3-5-7 Nim, with the winner taking the last stone. To review the Nim rules: Nim is a
game played with (unordered) piles of stones. Players alternate in taking one or more stones from a single
pile and discarding them. The player that takes the last stone from the board wins. 1-3-5-7 Nim is the
variant where there are four initial piles, of size 1, 3, 5 and 7.

Acro-Nim has four specific rules that alter the basic Nim game.

Pass Rule: Anytime a player takes three or more stones from a pile, the player receives a pass token. On
any turn, a player with a pass token may turn in a pass token rather than taking any stones. If both
players pass in succession, the game is terminated and scored a draw.

Poison Stone: One of the stones in the initial pile of seven stones is the poison stone. Any player taking
the poison stone, without taking at least one other stone from the pile, immediately loses the game
(even if the poison stone is the last available stone).

Equalizer Rule: Once per game, a player may on their turn choose to equalize the piles rather than taking
any stones. To equalize the piles, the player rearranges the stones in the piles so that all piles are as
even in size as possible. If the poison stone is present, and the arrangement is uneven, the poison stone
should be part of one of the larger piles.

Split Rule: Once per game, a player may on their turn choose to split a pile of two or more stones rather
than taking any stones. To split a pile, it is split into two piles, with the stones divided as evenly as
possible. A pile containing the poison stone may not be split.

2 A Z Specification Of Acro-Nim

The Z formal specification notation is described elsewhere. In this section, it will be used to formalize the
rules of Acro-Nim.

The state of a game of Acro-Nim consists of the state of the board and of two players. We will call these
players North and South, and let South move first.

The Acro-Nim board consists of piles of stones. We will model these piles using schema. There are no empty
piles: a pile ceases to exist when its last stone is removed. Our board design urges to keep track of whether
a pile contains poison stones or not as part of the pile state. While the rules given above mention only one
poison stone, it is convenient to generalize the specification to handle other initial positions.

Pile
stones : N1

poisoned : N
poisoned ≤ stones

The board is a set of piles.

Board
piles : P Pile

The initial board is the 1-3-5-7 board with the 7 pile containing one poisoned stone, described above. A
helper function is useful to generate the piles.

make pile : N1 × N→ Pile

∀n : N1; np : N; p : Pile •
make pile(n,np) = p ⇔

p.stones = n ∧ p.poisoned = np

InitBoard
Board

piles = {make pile(1, 0),make pile(3, 0),
make pile(5, 0),make pile(7, 1)}

The player state consists of the set of resources available to the player. While the game as described above
only allows a single equalizer or split, it is convenient to generalize the specification to allow other initial
states and/or rules.

Player
equalizes, splits, passes : N

The initial player state is as described.

InitPlayer
Player

equalizes = 1
splits = 1
passes = 0

It will be necessary to have a record of move types, and of the state of play. It is also necessary to name the
players.

MOVE ::= move take〈〈Pile × N1 × N〉〉 |
move pass | move split〈〈Pile〉〉 |
move equalize | move none

PLAYER ::= north | south
OPPONENT == {north 7→ south, south 7→ north}
PLAY ::= game continues | game won〈〈PLAYER〉〉 | game drawn

The game state consists of the board state, the state of each player, a record of the last move type (for the
draw rule), and an indication of the state of play.

GameState
Board
player : PLAYER → Player
to move : PLAYER
last move : MOVE
play : PLAY

The initial state is as described above.

InitGameState
GameState

InitBoard
ran player = InitPlayer
to move = south
last move = move none
play = game continues

The moves in Acro-Nim are all transformations on the game state. The player on move alternates.

Move
∆GameState

play = game continues
to move ′ = OPPONENT to move
player ′ to move ′ = player to move ′

There are four distinct types of move.

To conveniently annotate the various types of moves, it is useful to formalize the notion of a player state
adjustment.

adjust player : Player → (N× N× N)→ Player

∀ p, p′ : Player ; ne,ns,np : N •
p′ = adjust player p (ne,ns,np) ⇔

(p′.equalizes = p.equalizes + ne ∧
p′.splits = p.splits + ns ∧
p′.passes = p.passes + np)

A take move takes a selected number of (poisoned and/or non-poisoned) stones from a selected pile according
to the rules of Nim. A pile disappears when its last stone is taken. The poison stone has to be handled
carefully.

MoveTake
Move
p? : Pile
n?,np? : N
p?.stones ≥ n? ≥ 1
p?.poisoned ≥ np? ≥ n?− (p?.stones − p?.poisoned)
player ′ to move = if n? ≥ 3

then adjust player (player to move) (0, 0, 1)
else player to move

∃ rest : P Pile •
piles = rest ∪ {p?} ∧
piles ′ = rest∪

(if p?.stones = n?
then∅
else {make pile(p?.stones − n?, p?.poisoned − np?)})

play ′ = if n? = np?
then game won to move ′

else if piles ′ = ∅
then game won to move
else game continues

last move ′ = move take (p?,n?,np?)

If the player has a pass token, they may pass. If the previous move was a pass, then the game is drawn.
Otherwise, play continues in the same board state.

MovePass
Move

piles ′ = piles
(player ′ to move).passes > 0
player ′ to move = adjust player (player to move) (0, 0,−1)
play ′ = if last move = move pass

then game drawn
else game continues

last move ′ = move pass

A sensible way to handle split and equalizer moves is to construct a function that levels a set of piles. This
function proceeds in two phases. First, the set of piles is combined into a single big pile containing all the
stones.

pile up : P Pile → Pile

∀ p : Pile •
pile up {p} = p

∀ p, p′, q : Pile; ps : P Pile •
p′ = pile up ({p} ∪ ps) ⇔

(q = pile up ps ∧
p′.stones = p.stones + q .stones ∧
p′.poisoned = p.poisoned + q .poisoned)

Next, the big pile is split into the requested number of smaller piles. The piles are split as evenly as possible,
as are the poison stones: the poison stones are placed in the larger piles if possible.

level piles : N1 → P Pile → P Pile

∀ ps : P Pile •
level piles 1 ps = {pile up ps}

∀ p, p′, q : Pile; n,ns,np : N; ps, ps ′ : P Pile •
(n = #ps ′ ≥ 2 ∧
p = pile up ps ∧
p′.stones = p.stones div n ∧
p′.poisoned = p.poisoned div n ∧
q .stones = p.stones − p′.stones ∧
q .poisoned = p.poisoned − p′.poisoned ∧
ps ′ = {p′} ∪ level piles (n − 1) {q}) ⇒

ps ′ = level piles n ps

If the player has not consumed all of their splits, they may select a pile of two or more stones containing no
poison stones and split it as evenly as possible.

MoveSplit
Move
p? : Pile

p?.stones ≥ 2
p?.poisoned = 0
∃ rest : P Pile •

piles = rest ∪ {p?} ∧
piles ′ = rest ∪ level piles 2 {p?}

(player to move).splits > 0
player ′ to move = adjust player (player to move) (0,−1, 0)
last move ′ = move split p?

Finally, if the player has not consumed all of their equalizes, they may pile up all the stones and split them
as evenly as possible into the same number of piles they started with.

MoveEqualize
Move

piles ′ = level piles (#piles) piles
(player to move).equalizes > 0
player ′ to move = adjust player (player to move) (−1, 0, 0)
last move ′ = move equalize

A legal game of Acro-Nim consists simply of those moves described above that are legal in the given state.

InitGame =̂ InitGameState
Game =̂ MoveTake ∨ MovePass ∨

MoveSplit ∨ MoveEqualize

The terminal states are those which have no successor because the game is over.

